

Prise d'origine sur couple

pour variateurs série IMD / IMDL / IMDC

NOTICE

Principe pour effectuer une prise d'origine sur couple

1. Limiter le couple du moteur
2. Augmenter le paramètre d'erreur de poursuite maxi
3. Lancer un déplacement dans le sens souhaité, à petite vitesse
4. Attendre que l'erreur de poursuite augmente et atteigne un seuil
5. Remettre à zéro la position de l'axe
6. Dévalider l'axe
7. Remettre les valeurs initiales de couple moteur et du paramètre erreur de poursuite
8. Revalider l'axe

Exemple

Paramètres :

- Courant nominal = 2,1A
- Courant maxi = 200%
- Couple nominal = 2,5Nm
D'où un couple maxi de 5Nm (2,5N x 200%)
On souhaite limiter le couple à 0,63Nm lors de la prise d'origine, soit 12,6% du couple maxi
- Erreur de poursuite = 2 mm
- Précision des flottants = 0,01 (dans menu Options\Langage iDPL\Compilateur)

Programme :

Tâche 1

Prog

...

...

Run 4 'démarré la tâche de prise d'origine sur couple

Wait Status(4) = 0 'attente prise origine terminée

Vel% = 100

...

...

EndProg

Tâche 4

Prog

' **Mémorisation de l'erreur de poursuite initiale**

VL255 = ReadParam(60FBh,04h)

' **Forçage de l'erreur de poursuite à 10,00 mm (unité liée au paramètre Précision des flottants)**

WriteParam(60FBh,04h) = 1000

' **Mémorisation de la limitation de couple initiale**

VL254 = ReadParam(60F6h,10h)

' **Forçage de la limitation à 12,6 % du couple maxi (unité fixe exprimée en 0,1%)**

WriteParam(60F6h,10h) = 126

' **Lancement d'un déplacement à petite vitesse**

Vel% = 3

Stti -

' **Attente de passage du seuil d'erreur de poursuite (par exemple 5 mm)**

BOUCLE:

VR250 = Fe_S

If VR250 < 0 Then

 VR250 = -VR250

EndIf

If VR250 < 5 Goto BOUCLE ' seuil de 5 mm

' **Remise à zéro de la position de l'axe et dévalidation**

Home(0)

Axis Off

Delay 500

' **Restauration erreur de poursuite initiale**

WriteParam(60FBh,04h) = VL255

' **Restauration limitation de couple initiale**

WriteParam(60F6h,10h) = VL254

' **Asservissement de l'axe**

Axis On

Halt 4

EndProg

